

## Ü 425

### Wenn

- ✘ Wenn-Bedingung
- ✘ Benötigte Datei: Übungsmappe.ods
- ✓ Lernen Sie eine einfache Wenn-Bedingung kennen.

### Aufgabe

1. Öffnen Sie die **Übungsmappe.ods**. Wechseln Sie auf die Tabelle **Wenn**.
2. Wählen Sie die Zelle F6 aus.
3. Wenden Sie hier eine Funktion an, die folgendes bewirkt:  
Wenn die Summe des 1. Quartals über 3500 liegt, soll als Ergebnis ja, sonst nein erscheinen.
  - ▶ Geben Sie die Funktion ein =WENN(E6>3500;„ja“;„nein“)
  - ▶ WENN ist der Funktionsname
  - ▶ E6>3500 ist die Prüfung (das Kriterium)
  - ▶ „ja“ ist der Dann-Wert (Texte stehen dabei in Anführungszeichen)
  - ▶ „nein“ ist der Sonst-Wert
  - ▶ die 3 Argumente werden durch Strichpunkte getrennt
4. Bestätigen Sie den Eintrag und kopieren Sie die Funktion von F7 bis F12.
5. Wählen Sie G6 aus.
6. Erstellen Sie hier eine Wenn-Bedingung und verwenden Sie die Zelladressen aus G1 bis G3. Achten Sie auf die absoluten Bezüge.
7. Bestätigen Sie den Eintrag und kopieren Sie die Funktion von G7 bis G12.
8. Wählen Sie I6 aus.
9. Erstellen Sie hier eine Wenn-Bedingung und verwenden Sie die Zelladressen aus J1 bis J3. Achten Sie auf die absoluten Bezüge.
10. Bestätigen Sie den Eintrag und kopieren Sie die Funktion von I7 bis I12.
11. Wählen Sie J6 aus.
12. Erstellen Sie hier eine Wenn-Bedingung, verwenden Sie die Zelladressen aus J1 bis J3. Berechnen Sie diesmal gleich den Betrag. Achten Sie auf die absoluten Bezüge.
13. Bestätigen Sie den Eintrag und kopieren Sie die Funktion von J7 bis J12.

Beispiel

1. Geht es Ihnen gut? Mal gucken.

	A	B
1	Geht es Ihnen gut?	
2		
3		
4	Geben Sie in B1 Ja oder Nein ein Sie erhalten in B4 eine Information	

2. Geben Sie in B4 eine Funktion ein, die folgendes bewirkt:

Wenn der Eintrag in A1 Ja lautet, soll der Text erscheinen

*Gut! Dann können Sie heute ja länger bleiben und alles aufarbeiten!*

sonst soll der Text erscheinen

*Tja, nur die Harten kommen durch!*

Lösung:

=WENN(B1="JA";"Gut! Dann können Sie heute ja länger bleiben und alles aufarbeiten!";"Tja, nur die Harten kommen durch!")

Übung Umsätze

1. Öffnen Sie die **Übungsmappe.ods**. Fügen Sie nach der Tabelle **Wenn fertig** eine neue Tabelle ein.
2. Benennen Sie das Register um in **Wenn Umsatz**. Geben Sie hier die folgende Tabelle ein:

	A	B	C	D	E	F	G	H	I
1	<b>Umsätze der Vertreter/innen im 1. Quartal</b>								
2									
3									
4	Verkäufer/in	Klaus	Hofer	Haas	Forster	Summe Umsatz	Soll	+/-	Kommentar
5	Jänner	200	300	500	500		1600		
6	Februar	400	300	600	550		1800		
7	März	500	400	400	400		1600		
8									

3. Ermitteln Sie in die fehlenden Werte in den Spalten F und H.
4. Geben Sie in I5 eine Funktion ein, die folgendes bewirkt:  
Wenn der Inhalt der Zelle H5 kleiner 0 ist, soll „Abzug“ erschienen, sonst „Bonus“.
5. Bestätigen Sie den Eintrag und kopieren Sie die Funktion von I6 bis I7.

### Übung Abrechnung

1. Öffnen Sie die **Übungsmappe.ods**. Fügen Sie nach der Tabelle **Wenn Umsatz** eine neue Tabelle ein.
2. Benennen Sie das Register um in **Wenn Abrechnung**. Geben Sie hier die folgende Tabelle ein:

	A	B	C	D	E	F	G	H
1	<b>Abrechnung</b>							
2								
3	Preis je Stück	€ 9,90						
4	Bonus	€ 25,00						
5	ab Stückzahl	1.000						
6								
7	<i>Kundin/Kunde</i>	<i>1. Quartal</i>	<i>2. Quartal</i>	<i>3. Quartal</i>	<i>4. Quartal</i>	<i>Summe Anzahl</i>	<i>Summe Betrag</i>	<i>Bonus</i>
8	Kd 1	210	200	180	300			
9	Kd 2	240	360	320	280			
10	Kd 3	90	60	50	110			
11	Kd 4	420	190	270	390			
12	Kd 5	160	220	360	380			
13								

3. Ermitteln Sie in F8 *Summe Anzahl* mit der Summenfunktion.
4. Ermitteln Sie in G8 *Summe Betrag* mit der Formel *Summe Anzahl \* Preis je Stück* (also  $F8 * B3$ ).
5. Geben Sie in H8 eine Funktion ein, die folgendes bewirkt:  
Wenn F8 größer oder gleich 1000 ist, soll 25 erscheinen, sonst 0.
6. Bestätigen Sie den Eintrag und kopieren Sie die Funktion von H9 bis H12.
7. Fortgeschrittene ermitteln in I8 in einer Wenn-Bedingung gleich den aktuellen Abrechnungsbetrag.
8. Speichern Sie die Arbeitsmappe.

### Theorie

1. Erklären Sie den Aufbau der WENN-Funktion (Wenn-Bedingung).

### Antworten

## Aufgabe Wenn

1. Öffnen Sie die **Übungsmappe.ods**. Wechseln Sie auf die Tabelle **Wenn dann**.
2. Lösen Sie zum Einstieg die Bonusberechnung mit der Wenn-Funktion von C4 bis C11.
3. Die Kriterien finden Sie von F4 bis H5.
4. Lösen Sie anschließend die Bonusberechnung mit der Wenn-Funktion von C18 bis C25 und berechnen Sie in D18 bis D25 den Betrag.
5. Die Kriterien finden Sie von F18 bis H19.

## Exkurs Wenn verschachtelt

1. Jetzt verschachteln wir die Wenn-Funktion. Wir lösen die Bonusberechnung verschachtelt von C32 bis C39.
2. Die Kriterien finden Sie von F32 bis H34.
3. Verschachtelungen beginnen wir an einem Ende, zB mit dem Kriterium, dass der Umsatz aus B32 größer dem Betrag von 150 000 aus G34 ist.

**=WENN(B32>\$G\$34;**

4. Kommen wir zum Dann-Argument: Wenn das Kriterium zutrifft, wollen wir 15% aus H34 ausgeben. Das fügen wir in die Funktion ein, wir haben dann:

**=WENN(B32>\$G\$34;\$H\$34;**

5. Kommen wir zum Sonst-Argument. Da wir zwei weitere Bedingungen haben, verschachteln wir und beginnen mit einer neuen Wenn-Bedingung. Zunächst nur die neue Wenn-Funktion:

**WENN(B32>\$G\$33;\$H\$33;\$H\$32)**

6. Die komplette Funktion lautet schlussendlich:

**=WENN(B32>=\$G\$34;\$H\$34;WENN(B32>\$G\$33;\$H\$33;\$H\$32))**

7. Bitte beide Klammern schließen! Kopieren Sie die Funktion für die restlichen Zellen von C34 bis C39 und rechnen Sie den Betrag aus.

*Anmerkung: Beginnen Sie an einem Ende zu verschachteln und arbeiten Sie sich chronologisch zum anderen Ende der Bedingungen durch.*

## Übung

1. Erstellen Sie in einer neuen Mappe eine Rabattliste, gestaffelt je nach dem Auftragswert:

	A	B	C	D	E
1	<b>Verschachtelte Wenn-Bedingung</b>				
2					
3	<b>Auftragswert</b>	<b>Rabatt</b>		<b>Kriterien Auftragswert</b>	<b>Rabatt</b>
4	€ 850,00		unter	1000	kein Rabatt
5	€ 1.000,00		ab	1000	10%
6	€ 1.200,00		über	2000	15%
7	€ 2.800,00		über	5000	20%
8	€ 6.200,00		ab	10000	30%
9	€ 12.000,00				

2. Berechnen Sie den Rabatt in B4 und kopieren Sie die Funktion von B5 bis B9.
3. Formatieren Sie das Ergebnis im %-Format.

Lösung:

```
=WENN(A4>=D8;$E8;WENN(A4>D7;$E7;WENN(A4>D6;$E6;WENN(A4<=D5;$E5;E4))))
```

## Exkurs Wenn und

1. Von D46 bis D53 sollen zwei Kriterien zutreffen: Der Umsatz soll größer/gleich sein als 100 000 und der Umsatz höher/gleich sein als 45 000.
2. Dazu verwenden wir die *Wenn und* Bedingung:

**=WENN(UND(B46>=\$H\$46;C46>=\$H\$47);\$H\$48;\$H\$49)**

## Exkurs Wenn oder

1. Von D60 bis D67 soll eines der beiden Kriterien zutreffen: Der Umsatz soll entweder größer/gleich sein als 100 000 *oder* der Umsatz höher/gleich sein als 45 000.
2. Dazu verwenden wir die *Wenn oder* Bedingung:

**=WENN(ODER(B60>=\$H\$60;C60>=\$H\$61);\$H\$62;\$H\$63)**

## Exkurs Wenn und verschachtelt

1. Natürlich geht das auch verschachtelt:
2. Wir beginnen beim Verschachteln wieder an einem Ende: Ab 150 000 und einem Alter von 40 gibt es 15%.

**=WENN(UND(B73>=\$G\$75;C73>=\$G\$76);\$H\$75;**

3. Beim Sonst-Argument gibt es eine weitere *Wenn und* Bedingung: Liegt der Umsatz über 100 000 und das Alter ab 40, dann gibt es 10%, sonst 5%.

**WENN(UND(B73>\$G\$74;C73>=\$G\$76);\$H\$74;\$H\$73)**


4. Und das alles zusammen:

**=WENN(UND(B73>=\$G\$75;C73>=\$G\$76);\$H\$75;WENN(UND(B73>\$G\$74;C73>=\$G\$76);\$H\$74;\$H\$73))**

*Anmerkung: Das geht mit Wenn oder natürlich auch.*

## Exkurs Rechnung mit Wenn IST LEER

1. Öffnen Sie eine leere Arbeitsmappe.
2. Erstellen Sie zuerst die abgebildete Rechnung:

	A	B	C	D	E	F
1						
2						
3						
4						
5						
6						
7						
8	Kundennummer				Steuernummer	
9	Kundenname				Rechnungsdatum	
10	Kundenadresse				Anzahl	
11						
12	VerkäuferIn				Ust	
13						
14	<b>Position</b>	<b>Menge</b>	<b>Bezeichnung</b>	<b>Preis je Stk.</b>	<b>Betrag exkl.</b>	<b>Betrag inkl.</b>
15						
16						
17						
18						
19						
20						
21						
22						
23						
24						
25						
26						
27						
28						
29						
30			<b>Summen:</b>	€ -	€ -	€ -
31					Ust	€ -
32					Skonto	€ -
33	Netto zahlbar bis				Betrag	€ -
34						
35	Die Ware bleibt bis zur vollständigen Bezahlung unser Eigentum. Zahl- und klagbar in Vöcklabruck.					

3. Das heutige Rechnungsdatum in F9 geben Sie ein mit der Funktion **=HEUTE()**
4. Die Anzahl in F10 ermitteln Sie zB mit der Funktion **=ANZAHL(B16:B28)**
5. Geben Sie in F12 den aktuellen Umsatzsteuersatz ein.
6. In C33 geben Sie das heutige Datum und das 14-tägige Zahlungsziel ein mit der Funktion **=HEUTE()+14**
7. Ermitteln Sie in der Zeile 30 die Summen mit der Summenfunktion und formatieren Sie alle Zellen mit Preisen im €-Format.
8. Die Positionen sollen durchgezählt werden, aber nur soweit eine Menge in die Spalte B eingetragen wurde. Wir verwenden die Funktion Wenn IST LEER. Die komplette Funktion dazu lautet **=WENN(ISTLEER(B16);"";A15+1)**  
Kopieren Sie die Funktion von A17 bis A28. (Achten Sie auf die Formate.)
9. Für den Betrag exklusive erstellen Sie wieder eine Wenn IST LEER Funktion.

Die Lösung:

**=WENN(ISTLEER(B16);"";D16\*B16)**

10. Den Betrag inklusive berechnen Sie ebenso mit einer Wenn IST LEER Funktion.

Die Lösung:

**=WENN(ISTLEER(B16);"";E16\*\$F\$12+E16)**

11. Berechnen Sie in D32 den Skontosatz: Wenn der Rechnungsbetrag aus F30 ab 500 € beträgt, ist der Skonto 3%, ab 100 beträgt der Skonto 2%, sonst bleibt das Feld leer.

Die Lösung:

**=WENN(F30>=500;3%;WENN(F30>=100;2%;""))**

12. Weisen Sie in F31 die USt extra aus.

Die (einfache) Lösung:

**=F30-E30**

13. Ermitteln Sie den Skontobetrag in F32.

Die Lösung:

**=WENN(ISTLEER(F30);"";F30\*D32)**

14. Den Betrag in F32 ermitteln Sie aus Summe minus Skonto

Die Lösung:

**=F30-F32**

15. Es ist an der Zeit, die Rechnung auszuprobieren. Geben Sie eine Menge ein und einen Betrag – den Rest ermittelt Excel!

Zur Kontrolle finden Sie die Formelansicht auf der nächsten Seite.

	A	B	C	D	E	F
1	<h1>Rechnung</h1>					
2						
3						
4						
5						
6						
7						
8	Kundennummer				Steuernummer	
9	Kundenname				Rechnungsdatum	=HEUTE()
10	Kundenadresse				Anzahl	=ANZAHL(D16:D28)
11						
12	VerkäuferIn				Ust	0,2
13						
14						
		<b>Menge</b>	<b>Bezeichnung</b>	<b>Preis je Stk.</b>	<b>Betrag exkl.</b>	<b>Betrag inkl.</b>
16		=WENN(ISTLEER(B16);"";A15+1)			=WENN(ISTLEER(B16);"";D16*B16)	=WENN(ISTLEER(B16);"";E16*\$F\$12+E16)
17		=WENN(ISTLEER(B17);"";A16+1)			=WENN(ISTLEER(B17);"";D17*B17)	=WENN(ISTLEER(B17);"";E17*\$F\$12+E17)
18		=WENN(ISTLEER(B18);"";A17+1)			=WENN(ISTLEER(B18);"";D18*B18)	=WENN(ISTLEER(B18);"";E18*\$F\$12+E18)
19		=WENN(ISTLEER(B19);"";A18+1)			=WENN(ISTLEER(B19);"";D19*B19)	=WENN(ISTLEER(B19);"";E19*\$F\$12+E19)
20		=WENN(ISTLEER(B20);"";A19+1)			=WENN(ISTLEER(B20);"";D20*B20)	=WENN(ISTLEER(B20);"";E20*\$F\$12+E20)
21		=WENN(ISTLEER(B21);"";A20+1)			=WENN(ISTLEER(B21);"";D21*B21)	=WENN(ISTLEER(B21);"";E21*\$F\$12+E21)
22		=WENN(ISTLEER(B22);"";A21+1)			=WENN(ISTLEER(B22);"";D22*B22)	=WENN(ISTLEER(B22);"";E22*\$F\$12+E22)
23		=WENN(ISTLEER(B23);"";A22+1)			=WENN(ISTLEER(B23);"";D23*B23)	=WENN(ISTLEER(B23);"";E23*\$F\$12+E23)
24		=WENN(ISTLEER(B24);"";A23+1)			=WENN(ISTLEER(B24);"";D24*B24)	=WENN(ISTLEER(B24);"";E24*\$F\$12+E24)
25		=WENN(ISTLEER(B25);"";A24+1)			=WENN(ISTLEER(B25);"";D25*B25)	=WENN(ISTLEER(B25);"";E25*\$F\$12+E25)
26		=WENN(ISTLEER(B26);"";A25+1)			=WENN(ISTLEER(B26);"";D26*B26)	=WENN(ISTLEER(B26);"";E26*\$F\$12+E26)
27		=WENN(ISTLEER(B27);"";A26+1)			=WENN(ISTLEER(B27);"";D27*B27)	=WENN(ISTLEER(B27);"";E27*\$F\$12+E27)
28		=WENN(ISTLEER(B28);"";A27+1)			=WENN(ISTLEER(B28);"";D28*B28)	=WENN(ISTLEER(B28);"";E28*\$F\$12+E28)
29						
30		<b>Summen</b>	=SUMME(D16:D28)		=SUMME(E16:E28)	=SUMME(F16:F28)
31						Ust =F30-E30
32						Skonto =WENN(ISTLEER(F30);"";F30*D32)
33			=HEUTE()+14			Betrag =F30-F32
34						
35						

Die Ware bleibt bis zur vollständigen Bezahlung unser Eigentum. Zahl- und klagbar in Vöcklabruck.